

# **SAS/ETS<sup>®</sup> 14.2 User's Guide**

## **The LOAN Procedure**

This document is an individual chapter from *SAS/ETS® 14.2 User's Guide*.

The correct bibliographic citation for this manual is as follows: SAS Institute Inc. 2016. *SAS/ETS® 14.2 User's Guide*. Cary, NC: SAS Institute Inc.

#### **SAS/ETS® 14.2 User's Guide**

Copyright © 2016, SAS Institute Inc., Cary, NC, USA

All Rights Reserved. Produced in the United States of America.

**For a hard-copy book:** No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

**For a web download or e-book:** Your use of this publication shall be governed by the terms established by the vendor at the time you acquire this publication.

The scanning, uploading, and distribution of this book via the Internet or any other means without the permission of the publisher is illegal and punishable by law. Please purchase only authorized electronic editions and do not participate in or encourage electronic piracy of copyrighted materials. Your support of others' rights is appreciated.

**U.S. Government License Rights; Restricted Rights:** The Software and its documentation is commercial computer software developed at private expense and is provided with RESTRICTED RIGHTS to the United States Government. Use, duplication, or disclosure of the Software by the United States Government is subject to the license terms of this Agreement pursuant to, as applicable, FAR 12.212, DFAR 227.7202-1(a), DFAR 227.7202-3(a), and DFAR 227.7202-4, and, to the extent required under U.S. federal law, the minimum restricted rights as set out in FAR 52.227-19 (DEC 2007). If FAR 52.227-19 is applicable, this provision serves as notice under clause (c) thereof and no other notice is required to be affixed to the Software or documentation. The Government's rights in Software and documentation shall be only those set forth in this Agreement.

SAS Institute Inc., SAS Campus Drive, Cary, NC 27513-2414

November 2016

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

SAS software may be provided with certain third-party software, including but not limited to open-source software, which is licensed under its applicable third-party software license agreement. For license information about third-party software distributed with SAS software, refer to <http://support.sas.com/thirdpartylicenses>.

# Chapter 23

## The LOAN Procedure

### Contents

---

Overview: LOAN Procedure . . . . .	<b>1352</b>
Getting Started: LOAN Procedure . . . . .	<b>1352</b>
Analyzing Fixed Rate Loans . . . . .	1353
Analyzing Balloon Payment Loans . . . . .	1354
Analyzing Adjustable Rate Loans . . . . .	1355
Analyzing Buydown Rate Loans . . . . .	1356
Loan Repayment Schedule . . . . .	1357
Loan Comparison . . . . .	1358
Syntax: LOAN Procedure . . . . .	<b>1361</b>
Functional Summary . . . . .	1361
PROC LOAN Statement . . . . .	1363
ARM Statement . . . . .	1363
BALLOON Statement . . . . .	1366
BUYDOWN Statement . . . . .	1366
COMPARE Statement . . . . .	1366
FIXED Statement . . . . .	1368
Details: LOAN Procedure . . . . .	<b>1372</b>
Computational Details . . . . .	1372
Loan Comparison Details . . . . .	1374
OUT= Data Set . . . . .	1375
OUTCOMP= Data Set . . . . .	1375
OUTSUM= Data Set . . . . .	1376
Printed Output . . . . .	1377
ODS Table Names . . . . .	1378
Examples: LOAN Procedure . . . . .	<b>1379</b>
Example 23.1: Discount Points for Lower Interest Rates . . . . .	1379
Example 23.2: Refinancing a Loan . . . . .	1380
Example 23.3: Prepayments on a Loan . . . . .	1381
Example 23.4: Output Data Sets . . . . .	1383
Example 23.5: Piggyback Loans . . . . .	1384
References . . . . .	<b>1387</b>

---

---

## Overview: LOAN Procedure

The LOAN procedure analyzes and compares fixed rate, adjustable rate, buydown, and balloon payment loans. The LOAN procedure computes the loan parameters and outputs the loan summary information for each loan.

Multiple loan specifications can be processed and compared in terms of economic criteria such as after-tax or before-tax present worth of cost and true interest rate, breakeven of periodic payment and of interest paid, and outstanding balance at different periods in time. PROC LOAN selects the best alternative in terms of the specified economic criterion for each loan comparison period.

The LOAN procedure allows various payment and compounding intervals (including continuous compounding) and uniform or lump sum prepayments for a loan. Down payments, discount points, and other initialization costs can be included in the loan analysis and comparison.

The LOAN procedure does not support an input data set. All loans analyzed are specified with statements in the PROC LOAN step. The SAS DATA step provides a function MORT that can be used for data-driven analysis of many fixed-rate mortgage or installment loans. However, the MORT function supports only simple fixed rate loans.

---

## Getting Started: LOAN Procedure

PROC LOAN supports four types of loans. You specify each type of loan with the corresponding statement: FIXED, BALLOON, ARM, and BUYDOWN.

- **FIXED**—Fixed rate loans have a constant interest rate and periodic payment throughout the life of the loan.
- **BALLOON**—Balloon payment loans are fixed rate loans with lump sum payments in certain payment periods in addition to the constant periodic payment.
- **ARM**—Adjustable rate loans are those in which the interest rate and periodic payment vary over the life of the loan. The future interest rates of an adjustable rate loan are not known with certainty, but they will vary within specified limits according to terms stated in the loan agreement. In practice, the rate adjustment terms vary. PROC LOAN offers a flexible set of options to capture a wide variety of rate adjustment terms.
- **BUYDOWN**—Buydown rate loans are similar to adjustable rate loans, but the interest rate adjustments are predetermined at the initialization of the loan, usually by paying interest points at the time of loan initialization.

## Analyzing Fixed Rate Loans

The most common loan analysis is the calculation of the periodic payment when the loan amount, life, and interest rate are known. The following PROC LOAN statements analyze a 15-year (180 monthly payments) fixed rate loan for \$100,000 with an annual nominal interest rate of 7.5%:

```
proc loan;
    fixed amount=100000 rate=7.5 life=180;
run;
```

Another parameter the PROC LOAN statement can compute is the maximum amount you can borrow given the periodic payment you can afford and the rates available in the market. The following SAS statements analyze a loan for 180 monthly payments of \$900, with a nominal annual rate of 7.5%, and compute the maximum amount that can be borrowed:

```
proc loan;
    fixed payment=900 rate=7.5 life=180;
run;
```

Assume that you want to borrow \$100,000 and can pay \$900 a month. You know that the lender charges a 7.5% nominal interest rate compounded monthly. To determine how long it will take you to pay off your debt, use the following statements:

```
proc loan;
    fixed amount=100000 payment=900 rate=7.5;
run;
```

Sometimes, a loan is expressed in terms of the amount borrowed and the amount and number of periodic payments. In this case, you want to calculate the annual nominal rate charged on the loan to compare it to other alternatives. The following statements analyze a loan of \$100,000 paid in 180 monthly payments of \$800:

```
proc loan;
    fixed amount=100000 payment=800 life=180;
run;
```

There are four basic parameters that define a loan: life (number of periodic payments), principal amount, interest rate, and the periodic payment amount. PROC LOAN calculates the missing parameter among these four. Loan analysis output includes a loan summary table and an amortization schedule.

You can use the START= and LABEL= options to enhance your output. The START= option specifies the date of loan initialization and dates all the output accordingly. The LABEL= specification is used to label all output that corresponds to a particular loan; it is especially useful when multiple loans are analyzed. For example, the preceding statements for the first fixed rate loan are revised to include the START= and LABEL= options as follows:

```
proc loan start=1998:12;
    fixed amount=100000 rate=7.5 life=180
        label='BANK1, Fixed Rate';
run;
```

## Loan Summary Table

The loan summary table is produced by default and contains loan analysis information. It shows the principal amount, the costs at the time of loan initialization (down payment, discount points, and other loan initialization costs), the total payment and interest, the initial nominal and effective interest rates, payment and compounding intervals, the length of the loan in the time units specified, the start and end dates (if specified), a list of nominal and effective interest rates, and periodic payments throughout the life of the loan.

Figure 23.1 shows the loan summary table for the fixed rate loan labeled “BANK1, Fixed Rate.”

**Figure 23.1** Fixed Rate Loan Summary

The LOAN Procedure			
Fixed Rate Loan Summary			
BANK1, Fixed Rate			
Downpayment	0.00	Principal Amount	100000.00
Initialization	0.00	Points	0.00
Total Interest	66862.61	Nominal Rate	7.5000%
Total Payment	166862.61	Effective Rate	7.7633%
Pay Interval	MONTHLY	Compounding	MONTHLY
No. of Payments	180	No. of Compoundings	180
Start Date	DEC1998	End Date	DEC2013

  

Rates and Payments for BANK1, Fixed Rate			
Date	Nominal Rate	Effective Rate	Payment
DEC1998	7.5000%	7.7633%	927.01

The loan is initialized in December 1998 and paid off in December 2013. The monthly payment is calculated to be \$927.01, and the effective interest rate is 7.7633%. Over the 15 years, \$66,862.61 is paid for interest charges on the loan.

## Analyzing Balloon Payment Loans

You specify balloon payment loans like fixed rate loans, with the additional specification of the balloon payments. Assume you have an alternative to finance the \$100,000 investment with a 15-year balloon payment loan. The annual nominal rate is 7.5%, as in the fixed rate loan. The terms of the loan require two balloon payments of \$2000 and \$1000 at the 15th and 48th payment periods, respectively. These balloon payments keep the periodic payment lower than that of the fixed rate loan. The balloon payment loan is defined by the following BALLOON statement:

```
proc loan start=1998:12;
  balloon amount=100000 rate=7.5 life=180
    balloonpayment=(15=2000 48=1000)
    label = 'BANK2, with Balloon Payment';
run;
```

## List of Balloon Payments

In addition to the information for the fixed rate loan, the “Loan Summary Table” for the balloon payment loan includes a list of balloon payments in the list of rates and payments. For example, the balloon payment loan described previously includes two balloon payments, as shown in [Figure 23.2](#).

**Figure 23.2** List of Rates and Payments for a Balloon Payment Loan

The LOAN Procedure			
Rates and Payments for BANK2, with Balloon Payment			
Date	Nominal Rate	Effective Rate	Payment
DEC1998	7.5000%	7.7633%	903.25

  

Balloon Period	Payment
MAR2000	2000.00
DEC2002	1000.00

The periodic payment for the balloon payment loan is \$23.76 less than that of the fixed rate loan.

## Analyzing Adjustable Rate Loans

In addition to specifying the basic loan parameters, you need to specify the terms of the rate adjustments for an adjustable rate loan. There are many ways of stating the rate adjustment terms, and PROC LOAN facilitates all of them. For more information, see the section “[Rate Adjustment Terms Options](#)” on page 1364.

Assume that you have an alternative to finance the \$100,000 investment with a 15-year adjustable rate loan with an initial annual nominal interest rate of 5.5%. The rate adjustment terms specify a 0.5% annual cap, a 2.5% life cap, and a rate adjustment every 12 months. *Annual cap* refers to the maximum increase in interest rate per adjustment period, and *life cap* refers to the maximum increase over the life of the loan. The following ARM statement specifies this adjustable rate loan by assuming the interest rate adjustments will always increase by the maximum allowed by the terms of the loan. These assumptions are specified by the WORSTCASE and CAPS= options, as shown in the following statements:

```
proc loan start=1998:12;
  arm amount=100000 rate=5.5 life=180 worstcase
    caps=(0.5, 2.5)
    label='BANK3, Adjustable Rate';
run;
```

## List of Rates and Payments for Adjustable Rate Loans

The list of rates and payments in the loan summary table for the adjustable rate loans reflects the changes in the interest rates and payments and the dates these changes become effective. For the adjustable rate loan described previously, [Figure 23.3](#) shows the list of rates and payments that indicate five annual rate adjustments in addition to the initial rate and payment.

**Figure 23.3** List of Rates and Payments for an Adjustable Rate Loan

**The LOAN Procedure**

Rates and Payments for BANK3, Adjustable Rate			
Date	Nominal Rate	Effective Rate	Payment
DEC1998	5.5000%	5.6408%	817.08
JAN2000	6.0000%	6.1678%	842.33
JAN2001	6.5000%	6.6972%	866.44
JAN2002	7.0000%	7.2290%	889.32
JAN2003	7.5000%	7.7633%	910.88
JAN2004	8.0000%	8.3000%	931.03

Notice that the periodic payment of the adjustable rate loan as of January 2004 (\$931.03) exceeds that of the fixed rate loan (\$927.01).

## Analyzing Buydown Rate Loans

A 15-year buydown rate loan is another alternative to finance the \$100,000 investment. The nominal annual interest rate is 6.5% initially and will increase to 8% and 9% as of the 24th and 48th payment periods, respectively. The nominal annual interest rate is lower than that of the fixed rate alternative, at the cost of a 1% discount point (\$1000) paid at the initialization of the loan. The following BUYDOWN statement represents this loan alternative:

```
proc loan start=1998:12;
  buydown amount=100000 rate=6.5 life=180
    buydownrates=(24=8 48=9) pointpct=1
    label='BANK4, Buydown';
run;
```

## List of Rates and Payments for Buydown Rate Loans

Figure 23.4 shows the list of rates and payments in the loan summary table. It reflects the two rate adjustments and the corresponding monthly payments as well as the initial values for these parameters. As of December 2000, the periodic payment of the buydown loan exceeds the periodic payment for any of the other alternatives.

**Figure 23.4** List of Rates and Payments for a Buydown Rate Loan

**The LOAN Procedure**

Rates and Payments for BANK4, Buydown			
Date	Nominal Rate	Effective Rate	Payment
DEC1998	6.5000%	6.6972%	871.11
DEC2000	8.0000%	8.3000%	946.50
DEC2002	9.0000%	9.3807%	992.01



## Loan Repayment Schedule

In addition to the loan summary, you can print a loan repayment (amortization) schedule for each loan. For each payment period, this schedule contains the year and period within the year (or date, if the `START=` option is specified), the principal balance at the beginning of the period, the total payment, interest payment, principal repayment for the period, and the principal balance at the end of the period.

To print the first year of the amortization schedule for the fixed rate loan shown in Figure 23.5, use the following statements:

```
proc loan start=1998:12;
    fixed amount=100000 rate=7.5 life=180
        schedule=1
        label='BANK1, Fixed Rate';
run;
```

**Figure 23.5** Loan Repayment Schedule for the First Year  
**The LOAN Procedure**

Loan Repayment Schedule					
BANK1, Fixed Rate					
Date	Beginning Outstanding	Payment	Interest Payment	Principal Repayment	Ending Outstanding
DEC1998	100000.00	0.00	0.00	0.00	100000.00
DEC1998	100000.00	0.00	0.00	0.00	100000.00
JAN1999	100000.00	927.01	625.00	302.01	99697.99
FEB1999	99697.99	927.01	623.11	303.90	99394.09
MAR1999	99394.09	927.01	621.21	305.80	99088.29
APR1999	99088.29	927.01	619.30	307.71	98780.58
MAY1999	98780.58	927.01	617.38	309.63	98470.95
JUN1999	98470.95	927.01	615.44	311.57	98159.38
JUL1999	98159.38	927.01	613.50	313.51	97845.87
AUG1999	97845.87	927.01	611.54	315.47	97530.40
SEP1999	97530.40	927.01	609.57	317.44	97212.96
OCT1999	97212.96	927.01	607.58	319.43	96893.53
NOV1999	96893.53	927.01	605.58	321.43	96572.10
DEC1999	96572.10	927.01	603.58	323.43	96248.67
DEC1999	100000.00	11124.12	7372.79	3751.33	96248.67

The principal balance at the end of one year is \$96,248.67. The total payment for the year is \$11,124.12, of which \$3,751.33 went toward principal repayment.

You can also print the amortization schedule with annual summary information or for a specified number of years. The `SCHEDULE=YEARLY` option produces an annual summary loan amortization schedule, which is useful for loans with a long life. For example, to print the annual summary loan repayment schedule for the buydown loan shown in Figure 23.6, use the following statements:

```

proc loan start=1998:12;
    buydown amount=100000 rate=6.5 life=180
        buydownrates=(24=8 48=9) pointpct=1
        schedule=yearly
        label='BANK4, Buydown';
run;

```

**Figure 23.6** Annual Summary Loan Repayment Schedule**The LOAN Procedure**

Loan Repayment Schedule					
BANK4, Buydown					
Year	Beginning Outstanding	Interest Payment	Principal Repayment	Ending Outstanding	
1998	100000.00	1000.00	0.00	0.00	100000.00
1999	100000.00	10453.32	6380.07	4073.25	95926.75
2000	95926.75	10528.71	6222.21	4306.50	91620.25
2001	91620.25	11358.00	7178.57	4179.43	87440.82
2002	87440.82	11403.51	6901.12	4502.39	82938.43
2003	82938.43	11904.12	7276.64	4627.48	78310.95
2004	78310.95	11904.12	6842.58	5061.54	73249.41
2005	73249.41	11904.12	6367.76	5536.36	67713.05
2006	67713.05	11904.12	5848.43	6055.69	61657.36
2007	61657.36	11904.12	5280.35	6623.77	55033.59
2008	55033.59	11904.12	4659.00	7245.12	47788.47
2009	47788.47	11904.12	3979.34	7924.78	39863.69
2010	39863.69	11904.12	3235.96	8668.16	31195.53
2011	31195.53	11904.12	2422.83	9481.29	21714.24
2012	21714.24	11904.12	1533.41	10370.71	11343.53
2013	11343.53	11904.09	560.56	11343.53	0.00

**Loan Comparison**

The LOAN procedure can compare alternative loans on the basis of different economic criteria and help select the most desirable loan. You can compare alternative loans through different points in time. The economic criteria offered by PROC LOAN are as follows:

- outstanding principal balance—that is, the unpaid balance of the loan
- present worth of cost—that is, before-tax or after-tax net value of the loan cash flow through the comparison period. The cash flow includes all payments, discount points, initialization costs, down payment, and the outstanding principal balance at the comparison period.
- true interest rate—that is, before-tax or after-tax effective annual interest rate charged on the loan. The cash flow includes all payments, discount points, initialization costs, and the outstanding principal balance at the specified comparison period.
- periodic payment

- the total interest paid on the loan

The figures for present worth of cost, true interest rate, and interest paid are reported on the cash flow through the comparison period. The reported outstanding principal balance and the periodic payment are the values as of the comparison period.

The COMPARE statement specifies the type of comparison and the periods of comparison. For each period specified in the COMPARE statement, a loan comparison report is printed that also indicates the best alternative. Different criteria can lead to selection of different alternatives. Also, the period of comparison might change the desirable alternative. For more information, see the section “[Loan Comparison Details](#)” on page 1374.

## Comparison of 15-Year versus 30-Year Loan Alternatives

An issue that arises in the purchase of a house is the length of the loan life. Residential home loans are often for 15 or 30 years. Ordinarily, 15-year loans have a lower interest rate but higher periodic payments than 30-year loans. A comparison of both loans might identify the better loan for your means and needs. The following SAS statements compare two such loans:

```
proc loan start=1998:12 amount=120000;
    fixed rate=7.5 life=360 label='30 year loan';
    fixed rate=6.5 life=180 label='15 year loan';
    compare;
run;
```

### Default Loan Comparison Report

The default loan comparison report in [Figure 23.7](#) shows the ending outstanding balance, periodic payment, interest paid, and before-tax true rate at the end of 30 years. In the case of the default loan comparison, the selection of the best alternative is based on minimization of the true rate.

**Figure 23.7** Default Loan Comparison Report

The LOAN Procedure				
Loan Comparison Report				
Analysis through DEC2028				
Loan Label	Ending Outstanding	Payment	Interest Paid	True Rate
30 year loan	0.00	835.48	182058.02	7.76
15 year loan	0.00	1044.95	68159.02	6.70

**Note:** "15 year loan" is the best alternative based on true rate analysis through DEC2028.

Based on true rate, the best alternative is the 15-year loan. However, if the objective were to minimize the periodic payment, the 30-year loan would be the more desirable.

## Comparison of Fixed Rate and Adjustable Rate Loans

Suppose you want to compare a fixed rate loan to an adjustable rate alternative. The nominal interest rate on the adjustable rate loan is initially 1.5% lower than the fixed rate loan. The future rates of the adjustable rate loan are calculated using the worst-case scenario.

The interest paid on a loan might be deductible for tax purposes, depending on the purpose of the loan and applicable laws. In the following example, the TAXRATE=28 (income tax rate) option in the COMPARE statement bases the calculations of true interest rate on the after-tax cash flow. Assume, also, that you are uncertain as to how long you will keep this property. The AT=(60 120) option, as shown in the following example, produces two loan comparison reports through the end of the 5th and the 10th years, respectively:

```
proc loan start=1998:12 amount=120000 life=360;
  fixed rate=7.5 label='BANK1, Fixed Rate';
  arm   rate=6.0 worstcase caps=(0.5 2.5)
        label='BANK3, Adjustable Rate';
  compare taxrate=28 at=(60 120);
run;
```

### After-Tax Loan Comparison Reports

The two loan comparison reports in Figure 23.8 and Figure 23.9 show the ending outstanding balance, periodic payment, interest paid, and after-tax true rate at the end of five years and ten years, respectively.

**Figure 23.8** Loan Comparison Report as of December 2003

The LOAN Procedure				
Loan Comparison Report				
Analysis through DEC2003				
Loan Label	Ending Outstanding	Payment	Interest Paid	True Rate
BANK1, Fixed Rate	113540.74	839.06	43884.34	5.54
BANK3, Adjustable Rate	112958.49	871.83	40701.93	5.11

**Note:** "BANK3, Adjustable Rate" is the best alternative based on true rate analysis through DEC2003.

**Figure 23.9** Loan Comparison Report as of December 2008

Loan Comparison Report				
Analysis through DEC2008				
Loan Label	Ending Outstanding	Payment	Interest Paid	True Rate
BANK1, Fixed Rate	104153.49	839.06	84840.69	5.54
BANK3, Adjustable Rate	104810.98	909.57	87128.62	5.60

**Note:** "BANK1, Fixed Rate" is the best alternative based on true rate analysis through DEC2008.

The loan comparison report through December 2003 picks the adjustable rate loan as the best alternative, whereas the report through December 2008 shows the fixed rate loan as the better alternative. This implies that if you intend to keep the loan for 10 years or longer, the best alternative is the fixed rate alternative. Otherwise, the adjustable rate loan is the better alternative in spite of the worst-case scenario. Further analysis shows that the actual breakeven of true interest rate occurs at August 2008. That is, the desirable alternative switches from the adjustable rate loan to the fixed rate loan in August 2008.

Note that, under the assumption of worst-case scenario for the rate adjustments, the periodic payment for the adjustable rate loan already exceeds that of the fixed rate loan on December 2003 (as of the rate adjustment on January 2003 to be exact). If the objective were to minimize the periodic payment, the fixed rate loan would have been more desirable as of December 2003. However, all of the other criteria at that point still favor the adjustable rate loan.

## Syntax: LOAN Procedure

The following statements are used with PROC LOAN:

```
PROC LOAN options ;
  FIXED options ;
  BALLOON options ;
  ARM options ;
  BUYDOWN options ;
  COMPARE options ;
```

## Functional Summary

Table 23.1 summarizes the statements and options that control the LOAN procedure. Many of the loan specification options can be used on all of the statements except the COMPARE statement. For these options, the statement column is left blank. Options specific to a type of loan indicate the statement name.

**Table 23.1** Functional Summary

Description	Statement	Option
<b>Statements</b>		
Specify an adjustable rate loan	ARM	
Specify a balloon payment loan	BALLOON	
Specify a buydown rate loan	BUYDOWN	
Specify loan comparisons	COMPARE	
Specify a fixed rate loan	FIXED	
<b>Data Set Options</b>		
Specify output data set for loan summary	PROC LOAN	OUTSUM=
Specify output data set for repayment schedule		OUT=
Specify output data set for loan comparison	COMPARE	OUTCOMP=
<b>Printing Control Options</b>		
Suppress printing of loan summary report		NOSUMMARYPRINT
Suppress all printed output		NOPRINT

**Table 23.1** *continued*

Description	Statement	Option
Print amortization schedule		SCHEDULE=
Suppress printing of loan comparison report	COMPARE	NOCOMPRINT
<b>Required Specifications</b>		
Specify the loan amount		AMOUNT=
Specify life of loan as number of payments		LIFE=
Specify the periodic payment		PAYMENT=
Specify the initial annual nominal interest rate		RATE=
<b>Loan Specifications Options</b>		
Specify loan amount as percentage of price		AMOUNTPCT=
Specify time interval between compoundings		COMPOUND=
Specify down payment at loan initialization		DOWNPAYMENT=
Specify down payment as percentage of price		DOWNPAYPCT=
Specify amount paid for loan initialization		INITIAL=
Specify initialization costs as a percent		INITIALPCT=
Specify time interval between payments		INTERVAL=
Specify label for the loan		LABEL=
Specify amount paid for discount points		POINTS=
Specify discount points as a percent		POINTPCT=
Specify uniform or lump sum prepayments		PREPAYMENTS=
Specify the purchase price		PRICE=
Specify number of decimal places for rounding		ROUND=
Specify the date of loan initialization		START=
<b>Balloon Payment Loan Specification Option</b>		
Specify the list of balloon payments	BALLOON	BALLOONPAYMENT=
<b>Rate Adjustment Terms Options</b>		
Specify frequency of rate adjustments	ARM	ADJUSTFREQ=
Specify periodic and life cap on rate adjustment	ARM	CAPS=
Specify maximum rate adjustment	ARM	MAXADJUST=
Specify maximum annual nominal interest rate	ARM	MAXRATE=
Specify minimum annual nominal interest rate	ARM	MINRATE=
<b>Rate Adjustment Case Options</b>		
Specify best-case (optimistic) scenario	ARM	BESTCASE
Specify predicted interest rates	ARM	ESTIMATEDCASE=
Specify constant rate	ARM	FIXEDCASE
Specify worst-case (pessimistic) scenario	ARM	WORSTCASE
<b>Buydown Rate Loan Specification Option</b>		
Specify list of nominal interest rates	BUYDOWN	BUYDOWNRATES=

**Table 23.1** *continued*

Description	Statement	Option
<b>Loan Comparison Options</b>		
Specify all comparison criteria	COMPARE	ALL
Specify the loan comparison periods	COMPARE	AT=
Specify breakeven analysis of the interest paid	COMPARE	BREAKINTEREST
Specify breakeven analysis of periodic payment	COMPARE	BREAKPAYMENT
Specify minimum attractive rate of return	COMPARE	MARR=
Specify present worth of cost analysis	COMPARE	PWOF COST
Specify the income tax rate	COMPARE	TAXRATE=
Specify true interest rate analysis	COMPARE	TRUEINTEREST

## PROC LOAN Statement

### PROC LOAN *options* ;

The OUTSUM= option can be used in the PROC LOAN statement. In addition, the following loan specification options can be specified in the PROC LOAN statement to be used as defaults for all loans unless otherwise specified for a given loan:

AMOUNT=	INTERVAL=	POINTPCT=
AMOUNTPCT=	LABEL=	PREPAYMENTS=
COMPOUND=	LIFE=	PRICE=
DOWNPAYMENT=	NOSUMMARYPRINT	RATE=
DOWNPAYPCT=	NOPRINT	ROUND=
INITIAL=	PAYMENT=	START=
INITIALPCT=	POINTS=	SCHEDULE=

### Output Option

#### OUTSUM=SAS-data-set

creates an output data set that contains loan summary information for all loans other than those for which a different OUTSUM= output data set is specified.

## ARM Statement

### ARM *options* ;

The ARM statement specifies an adjustable rate loan where the future interest rates are not known with certainty but will vary within specified limits according to the terms stated in the loan agreement. In practice, the adjustment terms vary. Adjustments in the interest rate can be captured using the ARM statement options.

In addition to the required specifications and options listed under the FIXED statement, you can use the following options with the ARM statement.

## Rate Adjustment Terms Options

**ADJUSTFREQ=*n***

**ADF=*n***

specifies the number of periods, in terms of the INTERVAL= specification, between rate adjustments. INTERVAL=MONTH ADJUSTFREQ=6 indicates that the nominal interest rate can be adjusted every six months until the life cap or maximum rate (whichever is specified) is reached. The default is ADJUSTFREQ=12. The periodic payment is adjusted every adjustment period even if there is no rate change; therefore, if prepayments are made (as specified with the PREPAYMENTS= option), the periodic payment might change even if the nominal rate does not.

**CAPS=(*periodic-cap, life-cap*)**

specifies the maximum interest rate adjustment, in percent notation, allowed by the loan agreement. The *periodic cap* specifies the maximum adjustment allowed at each adjustment period. The *life cap* specifies the maximum total adjustment over the life of the loan. For example, a loan specified with CAPS=(0.5, 2) indicates that the nominal interest rate can change by 0.5% each adjustment period, and the annual nominal interest rate throughout the life of the loan will be within a 2% range of the initial annual nominal rate.

**MAXADJUST=*rate***

**MAXAD=*rate***

specifies the maximum rate adjustment, in percent notation, allowed at each adjustment period. Use the MAXADJUST= option with the MAXRATE= and MINRATE= options. The initial nominal rate plus the maximum adjustment should not exceed the specified MAXRATE= value. The initial nominal rate minus the maximum adjustment should not be less than the specified MINRATE= value.

**MAXRATE=*rate***

**MAXR=*rate***

specifies the maximum annual nominal rate, in percent notation, that might be charged on the loan. The maximum annual nominal rate should be greater than or equal to the initial annual nominal rate specified with the RATE= option.

**MINRATE=*rate***

**MINR=*rate***

specifies the minimum annual nominal rate, in percent notation, that might be charged on the loan. The minimum annual nominal rate should be less than or equal to the initial annual nominal rate specified with the RATE= option.

## Rate Adjustment Case Options

PROC LOAN supports four rate adjustment scenarios for analysis of adjustable rate loans: pessimistic (WORSTCASE), optimistic (BESTCASE), no-change (FIXEDCASE), and estimated (ESTIMATEDCASE). The estimated case enables you to analyze the adjustable rate loan with your predictions of future interest rates. The default is worst-case analysis. If more than one case is specified, worst-case analysis is performed. You can specify options for adjustable rate loans as follows:



**BESTCASE****B**

specifies a best-case analysis. The best-case analysis assumes that the interest rate charged on the loan will reach its minimum allowed limits at each adjustment period and over the life of the loan. If you use the BESTCASE option, you must specify either the CAPS= option or the MINRATE= and MAXADJUST= options.

**ESTIMATEDCASE=**( *date1=rate1 date2=rate2 ...* )

**ESTIMATEDCASE=**( *period1=rate1 period2=rate2 ...* )

**ESTC=**

specifies an estimated case analysis that indicates the rate adjustments will follow the rates you predict. This option specifies pairs of periods and estimated nominal interest rates.

The ESTIMATEDCASE= option can specify adjustments that cannot fit into the BESTCASE, WORSTCASE, or FIXEDCASE specifications, or “what-if” type analysis. If you specify the START= option, you can also specify the estimation periods as dates, in the form of SAS date literals. Estimated rates and the respective periods must be in time sequence.

If the estimated period falls between two adjustment periods (determined by ADJUSTFREQ= option), the rate is adjusted in the next adjustment period. The nominal interest rate charged on the loan is constant between two adjustment periods.

If any of the MAXRATE=, MINRATE=, CAPS=, and MAXADJUST= options are specified to indicate the rate adjustment terms of the loan agreement, these specifications are used to bound the rate adjustments. By using the ESTIMATEDCASE= option, you are predicting what the annual nominal rates in the market will be at different points in time, not necessarily the interest rate on your particular loan. For example, if the initial nominal rate (RATE= option) is 6.0, ADJUSTFREQ=6, MAXADJUST=0.5, and the ESTIMATEDCASE=(6=6.5, 12=7.5), the actual nominal rates charged on the loan would be 6.0% initially, 6.5% for the sixth through the eleventh periods, and 7.5% for the twelfth period onward.

**FIXEDCASE****FIXCASE**

specifies a fixed case analysis that assumes the rate will stay constant. The FIXEDCASE option calculates the ARM loan values similar to a fixed rate loan, but the payments are updated every adjustment period even if the rate does not change, leading to minor differences between the two methods. One such difference is in the way prepayments are handled. In a fixed rate loan, the rate and the payments are never adjusted; therefore, the payment stays the same over the life of the loan even when prepayments are made (instead, the life of the loan is shortened). In an ARM loan with the FIXEDCASE option, on the other hand, if prepayments are made, the payment is adjusted in the following adjustment period, leaving the life of the loan constant.

**WORSTCASE****W**

specifies a worst-case analysis. The worst-case analysis assumes that the interest rate charged on the loan will reach its maximum allowed limits at each rate adjustment period and over the life of the loan. If the WORSTCASE option is used, either the CAPS= option or the MAXRATE= and MAXADJUST= options must be specified.

---

## BALLOON Statement

**BALLOON** *options* ;

The BALLOON statement specifies a fixed rate loan with scheduled balloon payments in addition to the periodic payment. The following option is used in the BALLOON statement, in addition to the required options listed under the FIXED statement:

**BALLOONPAYMENT**=( *date1=payment1 date2=payment2 ...* )

**BALLOONPAYMENT**=( *period1=payment1 period2=payment2 ...* )

**BPAY**=( *date1=payment1 date2=payment2 ...* )

**BPAY**=( *period1=payment1 period2=payment2 ...* )

specifies pairs of periods and amounts of balloon (lump sum) payments in excess of the periodic payment during the life of the loan. You can also specify the balloon periods as dates if you specify the **START=** option. The dates are specified as SAS date literals. For example, **BALLOONPAYMENT= ( '1MAR2011' D=1000 )** specifies a payment of 1000 in March 2011.

If you do not specify this option, the calculations are identical to a loan specified in a **FIXED** statement. Balloon periods (or dates) and the respective balloon payments must be in time sequence.

---

## BUYDOWN Statement

**BUYDOWN** *options* ;

The BUYDOWN statement specifies a buydown rate loan. The buydown rate loans are similar to ARM loans, but the interest rate adjustments are predetermined at the initialization of the loan, usually by paying interest points at the time of loan initialization.

You must use all the required specifications and options listed under the **FIXED** statement with the **BUYDOWN** statement. The following option is specific to the **BUYDOWN** statement and is required:

**BUYDOWNRATES**=( *date1=rate1 date2=rate2 ...* )

**BUYDOWNRATES**=( *period1=rate1 period2=rate2 ...* )

**BDR=**

specifies pairs of periods and the predetermined nominal interest rates that will be charged on the loan starting at the corresponding time periods.

You can also specify the buydown periods as dates in the form of SAS date literals if you also specify the date of the initial payment by using a date value in the **START=** option. Buydown periods (or dates) and the respective buydown rates must be in time sequence.

---

## COMPARE Statement

**COMPARE** *options* ;

The **COMPARE** statement compares multiple loans, or it can be used with a single loan. You can use only one **COMPARE** statement. **COMPARE** statement options specify the periods and desired types of analysis

for loan comparison. The default analysis reports the outstanding principal balance, breakeven of payment, breakeven of interest paid, and before-tax true interest rate. The default comparison period corresponds to the first LIFE= option specification. If the LIFE= option is not specified for any loan, the loan comparison period defaults to the first calculated life.

You can use the following options with the COMPARE statement. For more information about loan comparison, see the section “[Loan Comparison Details](#)” on page 1374.

## Analysis Options

### ALL

is equivalent to specifying the BREAKINTEREST, BREAKPAYMENT, PWOF COST, and TRUEINTEREST options. The loan comparison report includes all the criteria. You need to specify the MARR= option for present worth of cost calculation.

**AT=( date1 date2 ... )**

**AT=( period1 period2 ... )**

specifies the periods for loan comparison reports. If you specify the START= option in the PROC LOAN statement, you can specify the AT= option as a list of dates expressed as SAS date literals instead of periods. The comparison periods do not need to be in time sequence. If you do not specify the AT= option, the comparison period defaults to the first LIFE= option specification. If you do not specify the LIFE= option for any of the loans, the loan comparison period defaults to the first calculated life.

### BREAKINTEREST

#### BI

specifies breakeven analysis of the interest paid. The loan comparison report includes the interest paid for each loan through the specified comparison period (AT= option).

### BREAKPAYMENT

#### BP

specifies breakeven analysis of payment. The periodic payment for each loan is reported for every comparison period specified in the AT=option.

**MARR=rate**

specifies the MARR (minimum attractive rate of return) in percent notation. The MARR reflects the cost of capital or the opportunity cost of money. The MARR= option is used in calculating the present worth of cost.

### PWOF COST

#### PWC

calculates the present worth of cost (net present value of costs) for each loan based on the cash flow through the specified comparison periods. The calculations account for down payment, initialization costs, and discount points, as well as the payments and outstanding principal balance at the comparison period. If you specify the TAXRATE= option, the present worth of cost is based on after-tax cash flow. Otherwise, before-tax present worth of cost is calculated. You need to specify the MARR= option for present worth of cost calculations.

**TAXRATE=***rate***TAX=***rate*

specifies income tax rate in percent notation for the after-tax calculations of the true interest rate and present worth of cost for those assets that qualify for tax deduction. If you specify this option, the amount specified in the POINTS= option and the interest paid on the loan are assumed to be tax-deductible. Otherwise, it is assumed that the asset does not qualify for tax deductions, and the cash flow is not adjusted for tax savings.

**TRUEINTEREST****TI**

calculates the true interest rate (effective interest rate based on the cash flow of all payments, initial-ization costs, discount points, and the outstanding principal balance at the comparison period) for all the specified loans through each comparison period. If you specify the TAXRATE= option, the true interest rate is based on after-tax cash flow. Otherwise, the before-tax true interest rate is calculated.

## Output Options

**NOCOMPRINT****NOCF**

suppresses the printing of the loan comparison report. The NOCOMPRINT option is usually used when an OUTCOMP= data set is created to store loan comparison information.

**OUTCOMP=***SAS-data-set*

writes the loan comparison report to an output data set.

---

## FIXED Statement

**FIXED** *options* ;

The FIXED statement specifies a fixed rate and periodic payment loan. It can be specified using the options that are common to all loan statements. The FIXED statement options are listed in this section.

You must specify three of the following options in each loan statement: AMOUNT=, LIFE=, RATE=, and PAYMENT=. The LOAN procedure calculates the fourth parameter based on the values you give the other three. If you specify all four of the options, the PAYMENT= specification is ignored, and the periodic payment is recalculated for consistency.

As an alternative to specifying the AMOUNT= option, you can specify the PRICE= option along with one of the following options to facilitate the calculation of the loan amount: AMOUNTPCT=, DOWNPAYMENT=, or DOWNPAYPCT=.

## Required Specifications

**AMOUNT=***amount***A=***amount*

specifies the loan amount (the outstanding principal balance at the initialization of the loan).

**LIFE=*n*****L=*n***

gives the life of the loan in number of payments. (The payment frequency is specified by the INTERVAL= option.) For example, if the life of the loan is 10 years with monthly payments, use LIFE=120 and INTERVAL=MONTH (default) to indicate a 10-year loan in which 120 monthly payments are made.

**PAYMENT=*amount*****P=*amount***

specifies the periodic payment. For ARM and BUYDOWN loans where the periodic payment might change, the PAYMENT= option specifies the initial amount of the periodic payment.

**RATE=*rate*****R=*rate***

specifies the initial annual (nominal) interest rate in percent notation. The rate specified must be in the range 0% to 120%. For example, use RATE=12.75 for a 12.75% loan. For ARM and BUYDOWN loans, where the rate might change over the life of the loan, the RATE= option specifies the initial annual interest rate.

## Specification Options

**AMOUNTPCT=*value*****APCT=*value***

specifies the loan amount as a percentage of the purchase price (PRICE= option). The AMOUNTPCT= specification is used to calculate the loan amount if the AMOUNT= option is not specified. The value specified must be in the range 1% to 100%.

If both the AMOUNTPCT= and DOWNPAYPCT= options are specified and the sum of their values is not equal to 100, the value of the down payment percentage is set equal to 100 minus the value of the amount percentage.

**COMPOUND=*time-unit***

specifies the time interval between compoundings. The default is the time unit given by the INTERVAL= option. If the INTERVAL= option is not used, then the default is COMPOUND=MONTH. The following time units are valid COMPOUND= values: CONTINUOUS, DAY, SEMIMONTH, MONTH, QUARTER, SEMIYEAR, and YEAR. The compounding interval is used to calculate the simple interest rate per payment period from the nominal annual interest rate or vice versa.

**DOWNPAYMENT=*amount*****DP=*amount***

specifies the down payment at the initialization of the loan. The down payment is included in the calculation of the present worth of cost but not in the calculation of the true interest rate. The after-tax analysis assumes that the down payment is not tax-deductible. (Specify after-tax analysis with the TAXRATE= option in the COMPARE statement.)

**DOWNPAYPCT=***value*

**DPCT=***value*

specifies the down payment as a percentage of the purchase price (PRICE= option). The DOWNPAYPCT= specification is used to calculate the down payment amount if you do not specify the DOWNPAYMENT= option. The value you specify must be in the range 0% to 99%.

If you specified both the AMOUNTPCT= and DOWNPAYPCT= options and the sum of their values is not equal to 100, the value of the down payment percentage is set equal to 100 minus the value of the amount percentage.

**INITIAL=***amount*

**INIT=***amount*

specifies the amount paid for loan initialization other than the discount points and down payment. This amount is included in the calculation of the present worth of cost and the true interest rate. The after-tax analysis assumes that the initial amount is not tax-deductible. (After-tax analysis is specified by the TAXRATE= option in the COMPARE statement.)

**INITIALPCT=***value*

**INITPCT=***value*

specifies the initialization costs as a percentage of the loan amount (AMOUNT= option). The INITIALPCT= specification is used to calculate the amount paid for loan initialization if you do not specify the INITIAL= option. The value you specify must be in the range of 0% to 100%.

**INTERVAL=***time-unit*

gives the time interval between periodic payments. The default is INTERVAL=MONTH. The following time units are valid INTERVAL values: SEMIMONTH, MONTH, QUARTER, SEMIYEAR, and YEAR.

**LABEL=***'loan-label'*

specifies a label for the loan. If you specify the LABEL= option, all output related to the loan is labeled accordingly. If you do not specify the LABEL= option, the loan is labeled by sequence number.

**POINTS=***amount*

**PNT=***amount*

specifies the amount paid for discount points at the initialization of the loan. This amount is included in the calculation of the present worth of cost and true interest rate. The amount paid for discount points is assumed to be tax-deductible in after-tax analysis (that is, if the TAXRATE= option is specified in the COMPARE statement).

**POINTPCT=***value*

**PNTPCT=***value*

specifies the discount points as a percentage of the loan amount (AMOUNT= option). The POINTPCT= specification is used to calculate the amount paid for discount points if you do not specify the POINTS= option. The value you specify must be in the range of 0% to 100%.

**PREPAYMENTS**=*amount*

**PREPAYMENTS**=( *date1=prepayment1 date2=prepayment2 ...* )

**PREPAYMENTS**=( *period1=prepayment1 period2=prepayment2 ...* )

**PREP**=

specifies either a uniform prepayment  $p$  throughout the life of the loan or lump sum prepayments. A uniform prepayment  $p$  is assumed to be paid with each periodic payment. Specify lump sum prepayments by pairs of periods (or dates) and respective prepayment amounts.

You can specify the prepayment periods as dates if you specify the **START**= option. Prepayment periods or dates and the respective prepayment amounts must be in time sequence. The prepayments are treated as principal payments, and the outstanding principal balance is adjusted accordingly. In the adjustable rate and buydown rate loans, if there is a rate adjustment after prepayments, the adjusted periodic payment is calculated based on the outstanding principal balance. The prepayments do not result in periodic payment amount adjustments in fixed rate and balloon payment loans.

**PRICE**=*amount*

**PRC**=*amount*

specifies the purchase price, which is the loan amount plus the down payment. If you specify the **PRICE**= option along with the loan amount (**AMOUNT**= option) or the down payment (**DOWNPAYMENT**= option), the value of the other one is calculated.

If you specify the **PRICE**= option with the **AMOUNTPCT**= or **DOWNPAYPCT**= options, the loan amount and the down payment are calculated.

**ROUND**= $n$

**ROUND**=NONE

specifies the number of decimal places to which the monetary amounts are rounded for the loan. Valid values for  $n$  are integers from 0 to 6. If you specify **ROUND**=NONE, the values are not rounded off internally, but the printed output is rounded off to two decimal places. The default is **ROUND**=2.

**START**=*SAS-date-literal*

**START**=*yyyy;period*

**S**=

gives the date of loan initialization. The first payment is assumed to be one payment interval after the start date. For example, you can specify the **START**= option as **START**=**'1APR2010'** **D** or as **START**=**2010:3**, where 3 is the third payment interval within the year 2010. If **INTERVAL**=**QUARTER**, 3 refers to the third quarter. If you specify the **START**= option, all output for the particular loan is dated accordingly.

## Output Options

**NOSUMMARYPRINT**

**NOSUMPR**

suppresses the printing of the loan summary report. The **NOSUMMARYPRINT** option is usually used when an **OUTSUM**= data set is created to store loan summary information.

**NOPRINT****NOP**

suppresses all printed output for the loan.

**OUT=SAS-data-set**

writes the loan amortization schedule to an output data set.

**OUTSUM=SAS-data-set**

writes the loan summary for the individual loan to an output data set.

**SCHEDULE****SCHEDULE=nyears****SCHEDULE=YEARLY****SCHED**

prints the amortization schedule for the loan. **SCHEDULE=nyears** specifies the number of years the printed amortization table covers. If you omit the number of years or specify a period longer than the loan life, the schedule is printed for the full term of the loan. **SCHEDULE=YEARLY** prints yearly summary information in the amortization schedule rather than the full amortization schedule. **SCHEDULE=YEARLY** is useful for long-term loans.

---

## Details: LOAN Procedure

---

### Computational Details

These terms are used in the formulas that follow:

$p$	periodic payment
$a$	principal amount
$r_a$	nominal annual rate
$f$	compounding frequency (per year)
$f'$	payment frequency (per year)
$r$	periodic rate
$r_e$	effective interest rate
$n$	total number of payments

The periodic rate, or the simple interest applied during a payment period, is given by

$$r = \left(1 + \frac{r_a}{f}\right)^{f/f'} - 1$$

Note that the interest calculation is performed at each payment period rather than at the compound period. This is done by adjusting the nominal rate. For more information, see Muksian (1984).



Note that when  $f = f'$  (that is, when the payment and compounding frequency coincide), the preceding expression reduces to the familiar form:

$$r = \frac{ra}{f}$$

The periodic rate for continuous compounding can be obtained from this general expression by taking the limit as the compounding frequency  $f$  goes to infinity. The resulting expression is

$$r = \exp\left(\frac{ra}{f'}\right) - 1$$

The effective interest rate, or annualized percentage rate (APR), is that rate which, if compounded once per year, is equivalent to the nominal annual rate compounded  $f$  times per year. Thus,

$$(1 + r_e) = (1 + r)^f = \left(1 + \frac{ra}{f}\right)^f$$

or

$$r_e = \left(1 + \frac{ra}{f}\right)^f - 1$$

For continuous compounding, the effective interest rate is given by

$$r_e = \exp(ra) - 1$$

For more information, see Muksian (1984).

The payment is calculated as

$$p = \frac{ar}{1 - \frac{1}{(1+r)^n}}$$

The amount is calculated as

$$a = \frac{p}{r} \left(1 - \frac{1}{(1+r)^n}\right)$$

Both the payment and amount are rounded to the nearest hundredth (cent) unless the ROUND= specification is different from the default, 2.

The total number of payments  $n$  is calculated as

$$n = \frac{-\ln\left(1 - \frac{ar}{p}\right)}{\ln(1+r)}$$

The total number of payments is rounded up to the nearest integer.

The nominal annual rate is calculated using the bisection method, with  $a$  as the objective and  $r$  starting in the interval between  $8 \times 10^{-6}$  and 0.1 with an initial midpoint 0.01 and successive midpoints bisecting.

---

## Loan Comparison Details

In order to compare the costs of different alternatives, the input cash flow for the alternatives must be represented in equivalent values. The equivalent value of a cash flow accounts for the time-value of money. That is, it is preferable to pay the same amount of money later than to pay it now, since the money can earn interest while you keep it. The MARR (minimum attractive rate of return) reflects the cost of capital or the opportunity cost of money—that is, the interest that would have been earned on the savings that is forgone by making the investment. The MARR is used to discount the cash flow of alternatives into equivalent values at a fixed point in time. The MARR can vary for each investor and for each investment. Therefore, the MARR= option must be specified in the COMPARE statement if present worth of cost (PWOF COST option) comparison is specified.

Present worth of cost reflects the equivalent amount at loan initialization of the loan cash flow discounted at MARR, not accounting for inflation. Present worth of cost accounts for the down payment, initialization costs, discount points, periodic payments, and the principal balance at the end of the report period. Therefore, it reflects the present worth of cost of the asset, not the loan. It is meaningful to use minimization of present worth of cost as a selection criterion only if the assets (down payment plus loan amount) are of the same value.

Another economic selection criterion is the rate of return (internal rate of return) of the alternatives. If interest is being earned by an alternative, the objective is to maximize the rate of return. If interest is being paid, as in loan alternatives, the best alternative is the one that minimizes the rate of return. The true interest rate reflects the effective annual rate charged on the loan based on the cash flow, including the initialization cost and the discount points.

The effects of taxes on different alternatives must be accounted for when these vary among different alternatives. Since interest costs on certain loans are tax-deductible, the comparisons for those loans are made based on the after-tax cash flows. The cost of the loan is reduced by the tax benefits it offers through the loan life if the TAXRATE= option is specified. The present worth of cost and true interest rate are calculated based on the after-tax cash flow of the loan. The down payment on the loan and initialization costs are assumed to be not tax-deductible in after-tax analysis. Discount points and the interest paid in each periodic payment are assumed to be tax-deductible if the TAXRATE= option is specified. If the TAXRATE= option is not specified, the present worth of cost and the true interest rate are based on before-tax cash flow, assuming that the interest paid on the specified loan does not qualify for tax benefits.

The other two selection criteria are breakeven analysis of periodic payment and interest paid. If the objective is to minimize the periodic payment, the best alternative is the one with the minimum periodic payment. If the objective is to minimize the interest paid on the principal, then the best alternative is the one with the least interest paid.

Another criterion might be the minimization of the outstanding balance of the loan at a particular point in time. For example, if you plan to sell a house before the end of the loan life (which is often the case), you might want to select the loan with the minimum principal balance at the time of the sale, since this balance must be paid at that time. The outstanding balance of the alternative loans is calculated for each loan comparison period by default.

If you specified the START= option in the PROC LOAN statement, the present worth of cost reflects the equivalent amount for each loan at that point in time. Any loan that has a START= specification different from the one in the PROC LOAN statement is not processed in the loan comparison.

The loan comparison report for each comparison period contains for each loan the loan label, outstanding balance, and any of the following measures if requested in the COMPARE statement: periodic payment (BREAKPAYMENT option), total interest paid to date (BREAKINTEREST option), present worth of cost (PWOF COST option), and true interest rate (TRUEINTEREST option). The best loan is selected on the basis of present worth of cost or true interest rate. If both PWOFCOST and TRUEINTEREST options are specified, present worth of cost is the basis for the selection of the best loan.

You can use the OUTCOMP= option in the COMPARE statement to write the loan comparison report to a data set. The NOCOMPRINT option suppresses the printing of a loan comparison report.

---

## OUT= Data Set

The OUT= option writes the loan amortization schedule to an output data set. The OUT= data set contains one observation for each payment period (or one observation for each year if you specified the SCHEDULE=YEARLY option). If you specified the START= option, the DATE variable denotes the date of the payment. Otherwise, YEAR and period variable (SEMIMONTH, MONTH, QUARTER, or SEMIYEAR) denote the payment year and period within the year.

The OUT= data set contains the following variables:

- DATE, date of the payment. DATE is included in the OUT= data set only when you specify the START= option.
- YEAR, year of the payment period. YEAR is included in the OUT= data set only when you do not specify the START= option.
- PERIOD, period within the year of the payment period. The name of the period variable matches the INTERVAL= specification (SEMIMONTH, MONTH, QUARTER, or SEMIYEAR.) The PERIOD variable is included in the OUT= data set only when you do not specify the START= option.
- BEGPRIN, beginning principal balance
- PAYMENT, payment
- INTEREST, interest payment
- PRIN, principal repayment
- ENDPRIN, ending principal balance

---

## OUTCOMP= Data Set

The OUTCOMP= option in the COMPARE statement writes the loan comparison analysis results to an output data set. If you specified the START= option, the DATE variable identifies the date of the loan comparison. Otherwise, the PERIOD variable identifies the comparison period.

The OUTCOMP= data set contains one observation for each loan and for each loan comparison period. The OUTCOMP= data set contains the following variables:

- DATE, date of loan comparison report. The DATE variable is included in the OUTCOMP= data set only when you specify the START= option.
- PERIOD, period of the loan comparison for the observation. The PERIOD variable is included in the OUTCOMP= data set only when you do not specify the START= option.
- LABEL, label string for the loan
- TYPE, type of the loan
- PAYMENT, periodic payment at the time of report. The PAYMENT is included in the OUTCOMP= data set if you specified the BREAKPAYMENT or ALL option or if you used default criteria.
- INTPAY, interest paid through the time of report. The INTPAY variable is included in the OUTCOMP= data set if you specified the BREAKINTEREST or ALL option or if you used default criteria.
- TRUERATE, true interest rate charged on the loan. The TRUERATE variable is included in the OUTCOMP= data set if you specified the TRUERATE or ALL option or if you used default criteria.
- PWOF COST, present worth of cost. The PWOF COST variable is included in the OUTCOMP= data set only if you specified the PWOF COST or ALL option.
- BALANCE, outstanding principal balance at the time of report

---

## OUTSUM= Data Set

The OUTSUM= option writes the loan summary to an output data set. If you specified this option in the PROC LOAN statement, the loan summary information for all loans is written to the specified data set, except for those loans for which you specified a different OUTSUM= data set in the ARM, BALLOON, BUYDOWN, or FIXED statement.

The OUTSUM= data set contains one observation for each loan and contains the following variables:

- TYPE, type of loan
- LABEL, loan label
- PAYMENT, periodic payment
- AMOUNT, loan principal
- DOWNPAY, down payment. DOWNPAY is included in the OUTSUM= data set only when you specify a down payment.
- INITIAL, loan initialization costs. INITIAL is included in the OUTSUM= data set only when you specify initialization costs.
- POINTS, discount points. POINTS is included in the OUTSUM= data set only when you specify discount points.

- TOTAL, total payment
- INTEREST, total interest paid
- RATE, nominal annual interest rate
- EFFRATE, effective interest rate
- INTERVAL, payment interval
- COMPOUND, compounding interval
- LIFE, loan life (that is, the number of payment intervals)
- NCOMPND, number of compounding intervals
- COMPUTE, computed loan parameter: life, amount, payment, or rate

If you specified the START= option either in the PROC LOAN statement or for the individual loan, the OUTSUM= data set also contains the following variables:

- BEGIN, start date
- END, loan termination date

---

## Printed Output

The output from PROC LOAN consists of the loan summary table, loan amortization schedule, and loan comparison report.

### Loan Summary Table

The loan summary table shows the total payment and interest, the initial nominal annual and effective interest rates, payment and compounding intervals, the length of the loan in the time units specified, the start and end dates if specified, a list of nominal and effective interest rates, and periodic payments throughout the life of the loan.

A list of balloon payments for balloon payment loans and a list of prepayments if specified are printed with their respective periods or dates.

The loan summary table is printed for each loan by default. The NOSUMMARYPRINT option specified in the PROC LOAN statement suppresses the printing of the loan summary table for all loans. The NOSUMMARYPRINT option can be specified in individual loan statements to selectively suppress the printing of the loan summary table.

## Loan Repayment Schedule

The amortization schedule contains for each payment period: the year and period within the year (or date, if you specified the `START=` option); principal balance at the beginning of the period; total payment, interest payment and principal payment for the period; and the principal balance at the end of the period. If you specified the `SCHEDULE=YEARLY` option, the amortization contains a summary for each year instead of for each payment period.

The amortization schedule is not printed by default. The `SCHEDULE` option in the `PROC LOAN` statement requests the printing of amortization tables for all loans. You can specify the `SCHEDULE` option in individual loan statements to selectively request the printing of the amortization schedule.

## Loan Comparison Report

The loan comparison report is processed for each report period and contains the results of economic analysis of the loans. The quantities reported can include the outstanding principal balance, after-tax or before-tax present worth of cost and true interest rate, periodic payment, and the interest paid through the report period for each loan. The best alternative is identified if the asset value (down payment plus loan amount) is the same for each alternative.

The loan comparison report is printed by default. The `NOCOMPRINT` option specified in the `COMPARE` statement suppresses the printing of the loan comparison report.

---

## ODS Table Names

`PROC LOAN` assigns a name to each table it creates. You can use these names to reference the table when using the Output Delivery System (ODS) to select tables and create output data sets. These names are listed in Table 23.2.

**Table 23.2** ODS Tables Produced in `PROC LOAN`

ODS Table Name	Description	Option
<b>ODS Tables Created by the <code>PROC LOAN</code>, <code>FIXED</code>, <code>ARM</code>, <code>BALLOON</code>, and <code>BUYDOWN</code> Statements</b>		
Repayment	Loan repayment schedule	<code>SCHEDULE</code>
<b>ODS Tables Created by the <code>FIXED</code>, <code>ARM</code>, <code>BALLOON</code>, and <code>BUYDOWN</code> Statements</b>		
LoanSummary	Loan summary	Default
RateList	Rates and payments	Default
PrepayList	Prepayments and periods	<code>PREPAYMENTS=</code>
<b>ODS Tables Created by the <code>BALLOON</code> Statement</b>		
BalloonList	Balloon payments and periods	Default
<b>ODS Tables Created by the <code>COMPARE</code> Statement</b>		
Comparison	Loan comparison report	Default

## Examples: LOAN Procedure

### Example 23.1: Discount Points for Lower Interest Rates

This example illustrates the comparison of two \$100,000 loans. The major difference between the two loans is that the nominal interest rate in the second loan is lower than the first with the added expense of paying discount points at the time of initialization.

Both alternatives are 30-year loans. The first loan is labeled “8.25% - no discount points” and the second one is labeled “8% - 1 discount point.”

Assume that the interest paid qualifies for a tax deduction and you are in the 33% tax bracket. Also, your minimum attractive rate of return (MARR) for an alternative investment is 4% (adjusted for tax rate).

You use the following statements to find the breakeven point in the life of the loan for your preference between the loans:

```
proc loan start=1992:1 nosummaryprint amount=100000 life=360;
  fixed rate=8.25 label='8.25% - no discount points';
  fixed rate=8 points=1000 label='8% - 1 discount point';
  compare at=(48 54 60) all taxrate=33 marr=4;
run;
```

Output 23.1.1 shows the loan comparison reports as of January 1996 (48th period), July 1996 (54th period), and January 1997 (60th period).

#### Output 23.1.1 Loan Comparison Reports for Discount Point Breakeven

##### The LOAN Procedure

Loan Comparison Report					
Analysis through JAN1996					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
8.25% - no discount points	96388.09	105546.17	751.27	32449.05	5.67
8% - 1 discount point	96219.32	105604.05	733.76	31439.80	5.69

**Note:** "8.25% - no discount points" is the best alternative based on present worth of cost analysis through JAN1996.

Loan Comparison Report					
Analysis through JUL1996					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
8.25% - no discount points	95847.27	106164.97	751.27	36415.85	5.67
8% - 1 discount point	95656.22	106153.97	733.76	35279.26	5.67

**Note:** "8% - 1 discount point" is the best alternative based on present worth of cost analysis through JUL1996.

**Output 23.1.1** *continued*


---

Loan Comparison Report					
Analysis through JAN1997					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
8.25% - no discount points	95283.74	106768.07	751.27	40359.94	5.67
8% - 1 discount point	95070.21	106689.80	733.76	39095.81	5.66

---

**Note:** "8% - 1 discount point" is the best alternative based on present worth of cost analysis through JAN1997.

Notice that the breakeven point for present worth of cost and true rate both happen on July 1996. This indicates that if you intend to keep the loan for 4.5 years or more, it is better to pay the discount points for the lower rate. If your objective is to minimize the interest paid or the periodic payment, the "8% - 1 discount point" loan is the preferred choice.

---

**Example 23.2: Refinancing a Loan**

Assume that you obtained a fixed rate 15-year loan in June 1995 for \$78,500 with a nominal annual rate of 9%. By early 1998, the market offers a 6.5% interest rate, and you are considering whether to refinance your loan.

Use the following statements to find out the status of the loan on February 1998. [Output 23.2.1](#) shows the results:

```
proc loan start=1995:6;
  fixed life=180 rate=9 amount=78500 noprint
    label='Original Loan';
  compare at=('10FEB1998'd);
run;
```

**Output 23.2.1** Loan Comparison Report for Original Loan

---

The LOAN Procedure				
Loan Comparison Report				
Analysis through FEB1998				
Loan Label	Ending Outstanding	Payment	Interest Paid	True Rate
Original Loan	71028.75	796.20	18007.15	9.38

---

The monthly payment on the original loan is \$796.20. The ending outstanding principal balance as of February is \$71,028.75. At this point, you might want to refinance your loan with another 15-year loan. The alternate loan has a 6.5% nominal annual rate. The initialization costs are \$1,419.00. Use the following statements to compare your alternatives:



```

proc loan start=1998:2 amount=71028.75;
  fixed rate=9 payment=796.20
    label='Keep the original loan' noprint;
  fixed life=180 rate=6.5 init=1419
    label='Refinance at 6.5%' noprint;
  compare at=(15 16) taxrate=33 marr=4 all;
run;

```

The comparison reports of May 1999 and June 1999 in [Output 23.2.2](#) illustrate the break even between the two alternatives. If you intend to keep the loan through June 1999 or longer, your initialization costs for the refinancing are justified. The periodic payment of the refinanced loan is \$618.74.

### Output 23.2.2 Loan Comparison Report for Refinancing Decision

#### The LOAN Procedure

---

##### Loan Comparison Report

##### Analysis through MAY1999

Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
Keep the original loan	66862.10	72737.27	796.20	7776.35	6.20
Refinance at 6.5%	67382.48	72747.51	618.74	5634.83	6.23

**Note:** "Keep the original loan" is the best alternative based on present worth of cost analysis through MAY1999.

---

##### Loan Comparison Report

##### Analysis through JUN1999

Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
Keep the original loan	66567.37	72844.52	796.20	8277.82	6.20
Refinance at 6.5%	67128.73	72766.42	618.74	5999.82	6.12

**Note:** "Refinance at 6.5%" is the best alternative based on present worth of cost analysis through JUN1999.

---

## Example 23.3: Prepayments on a Loan

This example compares a 30-year loan with and without prepayments. Assume the \$240,000 30-year loan has an 8.25% nominal annual rate. Use the following statements to see the effect of making uniform prepayments of \$500 with periodic payment:

```

proc loan start=1992:12 rate=8.25 amount=240000 life=360;
  fixed label='No prepayments';
  fixed label='With Prepayments' prepay=500;
  compare at=(120) taxrate=33 marr=4 all;
run;

```

[Output 23.3.1](#) through [Output 23.3.3](#) show the loan summary reports and the loan comparison report.

**Output 23.3.1** Loan Summary Reports without Prepayments**The LOAN Procedure**

Fixed Rate Loan Summary			
No prepayments			
Downpayment	0.00	Principal Amount	240000.00
Initialization	0.00	Points	0.00
Total Interest	409094.17	Nominal Rate	8.2500%
Total Payment	649094.17	Effective Rate	8.5692%
Pay Interval	MONTHLY	Compounding	MONTHLY
No. of Payments	360	No. of Compoundings	360
Start Date	DEC1992	End Date	DEC2022

  

Rates and Payments for No prepayments			
Date	Nominal Rate	Effective Rate	Payment
DEC1992	8.2500%	8.5692%	1803.04

**Output 23.3.2** Loan Summary Reports with Prepayments**The LOAN Procedure**

Fixed Rate Loan Summary			
With Prepayments			
Downpayment	0.00	Principal Amount	240000.00
Initialization	0.00	Points	0.00
Total Interest	183650.70	Nominal Rate	8.2500%
Total Payment	423650.70	Effective Rate	8.5692%
Pay Interval	MONTHLY	Compounding	MONTHLY
No. of Payments	184	No. of Compoundings	184
Start Date	DEC1992	End Date	APR2008

  

Rates and Payments for With Prepayments			
Date	Nominal Rate	Effective Rate	Payment
DEC1992	8.2500%	8.5692%	2303.04

**Output 23.3.3** Loan Comparison Report**The LOAN Procedure**

Loan Comparison Report					
Analysis through DEC2002					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
No prepayments	211608.05	268762.31	1803.04	187972.85	5.67
With Prepayments	118848.23	264149.25	2303.04	155213.03	5.67

**Note:** "With Prepayments" is the best alternative based on present worth of cost analysis through DEC2002.

Notice that with prepayments you pay off the loan in slightly more than 15 years. Also, the total payments and total interest are considerably lower with the prepayments. If you can afford the prepayments of \$500 each month, another alternative you should consider is using a 15-year loan, which is generally offered at a lower nominal interest rate.

---

## Example 23.4: Output Data Sets

This example shows the analysis and comparison of five alternative loans. Initialization cost, discount points, and both lump sum and periodic payments are included in the specification of these loans. Although no printed output is produced, the loan summary and loan comparison information is stored in the OUTSUM= and OUTCOMP= data sets.

```
proc loan start=1998:12 noprint outsum=loans
    amount=150000 life=360;

    fixed rate=7.5 life=180 prepayment=500
        label='BANK1, Fixed Rate';

    arm rate=5.5 estimatedcase=(12=7.5 18=8)
        label='BANK1, Adjustable Rate';

    buydown rate=7 interval=semimonth init=15000
        bdrates=(3=9 10=10) label='BANK2, Buydown';

    arm rate=5.75 worstcase caps=(0.5 2.5)
        adjustfreq=6 label='BANK3, Adjustable Rate'
        prepayments=(12=2000 36=5000);

    balloon rate=7.5 life=480
        points=1100 balloonpayment=(15=2000 48=2000)
        label='BANK4, with Balloon Payment';

    compare at=(120 360) all marr=7 tax=33 outcomp=comp;
run;

proc print data=loans;
run;

proc print data=comp;
run;
```

Output 23.4.1 and Output 23.4.2 illustrate the contents of the output data sets.

**Output 23.4.1** OUTSUM= Data Set

Obs	TYPE	LABEL	PAYMENT	AMOUNT	INITIAL	POINTS	TOTAL	INTEREST	RATE
1	FIXED	BANK1, Fixed Rate	1890.52	150000	0	0	207839.44	57839.44	0.0750
2	ARM	BANK1, Adjustable Rate	851.68	150000	0	0	390325.49	240325.49	0.0550
3	BUYDOWN	BANK2, Buydown	673.57	150000	15000	0	288858.08	138858.08	0.0700
4	ARM	BANK3, Adjustable Rate	875.36	150000	0	0	387647.82	237647.82	0.0575
5	BALLOON	BANK4, with Balloon Payment	965.36	150000	0	1100	467372.31	317372.31	0.0750

  

Obs	EFFRATE	INTERVAL	COMPOUND	LIFE	NCOMPND	COMPUTE	START	END
1	0.077633	MONTHLY	MONTHLY	110	110	PAYMENT	DEC1998	FEB2008
2	0.056408	MONTHLY	MONTHLY	360	360	PAYMENT	DEC1998	DEC2028
3	0.072399	SEMIMONTHLY	SEMIMONTHLY	360	360	PAYMENT	DEC1998	DEC2013
4	0.059040	MONTHLY	MONTHLY	360	360	PAYMENT	DEC1998	DEC2028
5	0.077633	MONTHLY	MONTHLY	480	480	PAYMENT	DEC1998	DEC2038

**Output 23.4.2** OUTCOMP= Data Set

Obs	DATE	TYPE	LABEL	PAYMENT	INTEREST	TRUERATE	PWOF COST	BALANCE
1	DEC2008	FIXED	BANK1, Fixed Rate	1772.76	57839.44	0.051424	137741.07	0.00
2	DEC2008	ARM	BANK1, Adjustable Rate	1093.97	108561.77	0.052212	130397.88	130788.65
3	DEC2008	BUYDOWN	BANK2, Buydown	803.98	118182.19	0.087784	161810.00	75798.19
4	DEC2008	ARM	BANK3, Adjustable Rate	1065.18	107015.58	0.053231	131955.90	125011.88
5	DEC2008	BALLOON	BANK4, with Balloon Payment	965.36	107906.61	0.052107	130242.56	138063.41
6	DEC2028	FIXED	BANK1, Fixed Rate	1772.76	57839.44	0.051424	137741.07	0.00
7	DEC2028	ARM	BANK1, Adjustable Rate	1094.01	240325.49	0.053247	121980.94	0.00
8	DEC2028	BUYDOWN	BANK2, Buydown	800.46	138858.08	0.086079	161536.44	0.00
9	DEC2028	ARM	BANK3, Adjustable Rate	1065.20	237647.82	0.054528	124700.22	0.00
10	DEC2028	BALLOON	BANK4, with Balloon Payment	965.36	282855.86	0.051800	117294.50	81326.26

## Example 23.5: Piggyback Loans

The *piggyback* loan is becoming a widely available alternative. Borrowers like to avoid the PMI (private mortgage insurance) required with loans where the borrower has a down payment of less than 20% of the price. The piggyback allows a secondary home equity loan to be packaged with a primary loan with less than 20% down payment. The secondary loan usually has a shorter life and higher interest rate. The interest paid on both loans are tax-deductible whereas PMI does not qualify for a tax deduction.

The following example compares a conventional fixed rate loan with 20% down as opposed to a piggyback loan: one primary fixed rate with 10% down payment and a secondary, home equity loan for 10% of the original price. All loans have monthly payments.

The conventional loan alternative is a 30-year loan with a fixed annual rate of 7.5%. The primary loan in the piggyback loan setup is also a 30-year loan with a fixed annual rate of 7.75%. The secondary loan is a 15-year loan with a fixed annual interest rate of 8.25%.

The comparison output for the two loans comprising the piggyback loan is aggregated using the TIMESERIES procedure with a minimum of specified options:

- The INTERVAL= option requests that the data be aggregated into periods of length 5 years beginning on the 25th month, resulting in appropriately identified periods.
- The ACC=TOTAL option specifies that the output should reflect accumulated totals as opposed to, say, averages.
- The NOTSORTED option indicates that the input data set has not been sorted by the ID variable.

For more information about this procedure, see Chapter 39, “[The TIMESERIES Procedure](#).”

Use the following statements to analyze the conventional loan, as well as the piggyback alternative, and compare them on the basis of their present worth of cost, outstanding balance, and interest payment amounts at the end of 5, 10, and 15 years into the loan life:

```

title1 'LOAN: Piggyback loan example';

title2 'LOAN: Conventional loan';

proc loan start=2002:1 noprint;

    fixed price=200000 dp=40000 rate=7.5 life=360
        label='20 percent down: Conventional Fixed Rate' ;

    compare at=(60 120 180) pwofcost taxrate=30 marr=12
        breakpay breakint outcomp=comploans;

run;

title2 'LOAN: Piggyback: Primary Loan';

proc loan start=2002:1 noprint;

    fixed amount=160000 dp=20000 rate=7.75 life=360
        label='Piggyback: Primary loan' out=loan1;

    compare at=(60 120 180 ) pwofcost taxrate=30 marr=12
        breakpay breakint outcomp=cloan1;

run;

title2 'LOAN: Piggyback: Secondary (Home Equity) Loan';

proc loan start=2002:1 noprint;

    fixed amount=20000 rate=8.25 life=180
        label='Piggyback: Secondary (Home Equity) Loan' out=loan2;

    compare at=(60 120 180 ) pwofcost taxrate=30 marr=12
        breakpay breakint outcomp=cloan2;

run;

data cloan12;
    set cloan1 cloan2;
run;
```

```

proc timeseries data=cloan12 out= totcomp ;
    id date interval=year5.25 acc=total notsorted;
    var payment interest pwofcost balance ;
run;

/*-- LOAN: Piggyback loan --*/
title;
proc print data=totcomp;
    format date monyy7.;
run;

data comploans;
    set comploans;
    drop type label;
run;

/*-- LOAN: Conventional Loan --*/
title;
proc print data=comploans;
run;

```

The loan comparisons in [Output 23.5.1](#) and [Output 23.5.2](#) illustrate the after-tax comparison of the loans. The after-tax present value of cost for the piggyback loan is lower than the 20% down conventional fixed rate loan.

**Output 23.5.1** Piggyback Loan

Obs	DATE	PAYMENT	INTEREST	PWOF COST	BALANCE
1	JAN2007	1340.29	67992.92	157157.41	167575.52
2	JAN2012	1340.29	129973.53	135556.98	149138.73
3	JAN2017	1339.66	183028.58	125285.77	121777.01

**Output 23.5.2** Conventional Loan

Obs	DATE	PAYMENT	INTEREST	PWOF COST	BALANCE
1	JAN2007	1118.74	58512.54	160436.81	151388.14
2	JAN2012	1118.74	113121.41	140081.64	138872.61
3	JAN2017	1118.74	162056.97	130014.97	120683.77

---

## References

- DeGarmo, E. P., Sullivan, W. G., and Canada, J. R. (1984). *Engineering Economy*. 7th ed. New York: Macmillan.
- Muksian, R. (1984). *Financial Mathematics Handbook*. Englewood Cliffs, NJ: Prentice-Hall.
- Newnan, D. G. (1988). *Engineering Economic Analysis*. 3rd ed. San Jose, CA: Engineering Press.
- Riggs, J. L., and West, T. M. (1986). *Essentials of Engineering Economics*. 2nd ed. New York: McGraw-Hill.

# Subject Index

adjustable rate mortgage, *see* LOAN procedure

    LOAN procedure, 1352

amortization schedule

    LOAN procedure, 1378

balloon payment mortgage, *see* LOAN procedure

    LOAN procedure, 1352

breakeven analysis

    LOAN procedure, 1374

buydown rate loans, *see* LOAN procedure

    LOAN procedure, 1352

comparing loans

    LOAN procedure, 1358, 1374, 1378

continuous compounding

    LOAN procedure, 1372

fixed rate mortgage, *see* LOAN procedure

    LOAN procedure, 1352

installment loans, *see* LOAN procedure

interest rates

    LOAN procedure, 1373

internal rate of return

    LOAN procedure, 1374

LOAN procedure

    adjustable rate mortgage, 1351, 1352

    amortization schedule, 1378

    balloon payment mortgage, 1351, 1352

    breakeven analysis, 1374

    buydown rate loans, 1351, 1352

    comparing loans, 1358, 1374, 1378

    continuous compounding, 1372

    fixed rate mortgage, 1351, 1352

    installment loans, 1351

    interest rates, 1373

    internal rate of return, 1374

    loan repayment schedule, 1378

    loan summary table, 1377

    loans analysis, 1351

    minimum attractive rate of return, 1374

    mortgage loans, 1351

    output data sets, 1375, 1376

    output table names, 1378

    present worth of cost, 1374

    rate adjustment cases, 1364

    taxes, 1374

    true interest rate, 1374

    types of loans, 1352

loan repayment schedule

    LOAN procedure, 1378

loan summary table

    LOAN procedure, 1377

loans analysis, *see* LOAN procedure

MARR, *see* minimum attractive rate of return

minimum attractive rate of return

    LOAN procedure, 1374

minimum attractive rate of return (MARR), 1374

mortgage loans, *see* LOAN procedure

output data sets

    LOAN procedure, 1375, 1376

output table names

    LOAN procedure, 1378

present worth of cost

    LOAN procedure, 1374

rate adjustment cases

    LOAN procedure, 1364

taxes

    LOAN procedure, 1374

true interest rate

    LOAN procedure, 1374

types of loans

    LOAN procedure, 1352





# Syntax Index

A= option  
    FIXED statement (LOAN), [1368](#)

ADF= option  
    ARM statement (LOAN), [1364](#)

ADJUSTFREQ= option  
    ARM statement (LOAN), [1364](#)

ALL option  
    COMPARE statement (LOAN), [1367](#)

AMOUNT= option  
    FIXED statement (LOAN), [1368](#)

AMOUNTPCT= option  
    FIXED statement (LOAN), [1369](#)

APCT= option  
    FIXED statement (LOAN), [1369](#)

ARM statement  
    LOAN procedure, [1363](#)

AT= option  
    COMPARE statement (LOAN), [1367](#)

B option  
    ARM statement (LOAN), [1365](#)

BALLOON statement  
    LOAN procedure, [1366](#)

BALLOONPAYMENT= option  
    BALLOON statement (LOAN), [1366](#)

BESTCASE option  
    ARM statement (LOAN), [1365](#)

BI option  
    COMPARE statement (LOAN), [1367](#)

BP option  
    COMPARE statement (LOAN), [1367](#)

BREAKINTEREST option  
    COMPARE statement (LOAN), [1367](#)

BREAKPAYMENT option  
    COMPARE statement (LOAN), [1367](#)

BUYDOWN statement  
    LOAN procedure, [1366](#)

BUYDOWNRATES= option  
    BUYDOWN statement (LOAN), [1366](#)

CAPS= option  
    ARM statement (LOAN), [1364](#)

COMPARE statement  
    LOAN procedure, [1366](#)

COMPOUND= option  
    FIXED statement (LOAN), [1369](#)

DOWNPAYMENT= option  
    FIXED statement (LOAN), [1369](#)

DOWNPAYPCT= option  
    FIXED statement (LOAN), [1370](#)

DP= option  
    FIXED statement (LOAN), [1369](#)

DPCT= option  
    FIXED statement (LOAN), [1370](#)

ESTIMATEDCASE= option  
    ARM statement (LOAN), [1365](#)

FIXED statement  
    LOAN procedure, [1368](#)

FIXEDCASE option  
    ARM statement (LOAN), [1365](#)

INIT= option  
    FIXED statement (LOAN), [1370](#)

INITIAL= option  
    FIXED statement (LOAN), [1370](#)

INITIALPCT= option  
    FIXED statement (LOAN), [1370](#)

INITPCT= option  
    FIXED statement (LOAN), [1370](#)

INTERVAL=option  
    FIXED statement (LOAN), [1370](#)

L= option  
    FIXED statement (LOAN), [1369](#)

LABEL= option  
    FIXED statement (LOAN), [1370](#)

LIFE= option  
    FIXED statement (LOAN), [1369](#)

LOAN procedure, [1361](#)  
    syntax, [1361](#)

MARR= option  
    COMPARE statement (LOAN), [1367](#)

MAXAD= option  
    ARM statement (LOAN), [1364](#)

MAXADJUST= option  
    ARM statement (LOAN), [1364](#)

MAXR= option  
    ARM statement (LOAN), [1364](#)

MAXRATE= option  
    ARM statement (LOAN), [1364](#)

MINR= option  
    ARM statement (LOAN), [1364](#)

MINRATE= option  
    ARM statement (LOAN), [1364](#)

NOCOMPRINT option  
    COMPARE statement (LOAN), 1368  
NOP option  
    FIXED statement (LOAN), 1372  
NOPRINT option  
    FIXED statement (LOAN), 1372  
NOSUMMARYPRINT option  
    FIXED statement (LOAN), 1371  
NOSUMPR option  
    FIXED statement (LOAN), 1371  
  
OUT= option  
    FIXED statement (LOAN), 1372, 1375  
OUTCOMP= option  
    COMPARE statement (LOAN), 1368, 1375  
OUTSUM= option  
    FIXED statement (LOAN), 1372  
    PROC LOAN statement, 1363, 1376  
  
P= option  
    FIXED statement (LOAN), 1369  
PAYMENT= option  
    FIXED statement (LOAN), 1369  
PNT= option  
    FIXED statement (LOAN), 1370  
PNTPCT= option  
    FIXED statement (LOAN), 1370  
POINTPCT= option  
    FIXED statement (LOAN), 1370  
POINTS= option  
    FIXED statement (LOAN), 1370  
PRC= option  
    FIXED statement (LOAN), 1371  
PREPAYMENTS= option  
    FIXED statement (LOAN), 1371  
PRICE= option  
    FIXED statement (LOAN), 1371  
PROC LOAN statement, 1363  
PWC option  
    COMPARE statement (LOAN), 1367  
PWOF COST option  
    COMPARE statement (LOAN), 1367  
  
R= option  
    FIXED statement (LOAN), 1369  
RATE= option  
    FIXED statement (LOAN), 1369  
ROUND= NONE option  
    FIXED statement (LOAN), 1371  
ROUND= option  
    FIXED statement (LOAN), 1371  
  
SCHEDULE option  
    FIXED statement (LOAN), 1372  
SCHEDULE= option

    FIXED statement (LOAN), 1372  
SCHEDULE= YEARLY option  
    FIXED statement (LOAN), 1372  
START= option  
    FIXED statement (LOAN), 1371  
  
TAX= option  
    COMPARE statement (LOAN), 1368  
TAXRATE= option  
    COMPARE statement (LOAN), 1368  
TI option  
    COMPARE statement (LOAN), 1368  
TRUEINTEREST option  
    COMPARE statement (LOAN), 1368  
  
W option  
    ARM statement (LOAN), 1365  
WORSTCASE option  
    ARM statement (LOAN), 1365